

G13DXF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G13DXF calculates the zeros of a vector autoregressive (or moving average) operator. This routine is likely to be used in conjunction with G05HDF, G13ASF, G13DCF or G13DSF.

2 Specification

```

SUBROUTINE G13DXF(K, IP, PAR, RR, RI, RMOD, WORK, IWORK, IFAIL)
  INTEGER          K, IP, IWORK(K*IP), IFAIL
  real            PAR(IP*K*K), RR(K*IP), RI(K*IP), RMOD(K*IP),
1                 WORK(K*K*IP*IP)

```

3 Description

Consider the vector autoregressive moving average (VARMA) model

$$W_t - \mu = \phi_1(W_{t-1} - \mu) + \phi_2(W_{t-2} - \mu) + \dots + \phi_p(W_{t-p} - \mu) + \epsilon_t - \theta_1\epsilon_{t-1} - \theta_2\epsilon_{t-2} - \dots - \theta_q\epsilon_{t-q} \quad (1)$$

where W_t denotes a vector of k time series and ϵ_t is a vector of k residual series having zero mean and a constant variance-covariance matrix. The components of ϵ_t are also assumed to be uncorrelated at non-simultaneous lags. $\phi_1, \phi_2, \dots, \phi_p$, denotes a sequence of k by k matrices of autoregressive (AR) parameters and $\theta_1, \theta_2, \dots, \theta_q$, denotes a sequence of k by k matrices of moving average (MA) parameters. μ is a vector of length k containing the series means. Let

$$A(\phi) = \begin{bmatrix} \phi_1 & I & 0 & \dots & \dots & \dots & 0 \\ \phi_2 & 0 & I & 0 & \dots & \dots & 0 \\ \vdots & & & & & & \\ \vdots & & & & & & \\ \phi_{p-1} & 0 & \dots & \dots & \dots & 0 & I \\ \phi_p & 0 & \dots & \dots & \dots & 0 & 0 \end{bmatrix}_{pk \times pk}$$

where I denotes the k by k identity matrix.

The model (1) is said to be stationary if the eigenvalues of $A(\phi)$ lie inside the unit circle. Similarly let

$$B(\theta) = \begin{bmatrix} \theta_1 & I & 0 & \dots & \dots & \dots & 0 \\ \theta_2 & 0 & I & 0 & \dots & \dots & 0 \\ \vdots & & & & & & \\ \vdots & & & & & & \\ \vdots & & & & & & \\ \theta_{q-1} & 0 & \dots & \dots & \dots & 0 & I \\ \theta_q & 0 & \dots & \dots & \dots & 0 & 0 \end{bmatrix}_{qk \times qk}$$

Then the model is said to be invertible if the eigenvalues of $B(\theta)$ lie inside the unit circle.

G13DXF returns the pk eigenvalues of $A(\phi)$ (or the qk eigenvalues of $B(\theta)$) along with their moduli, in descending order of magnitude. Thus to check for stationarity or invertibility the user should check whether the modulus of the largest eigenvalue is less than one.

4 References

- [1] Wei W W S (1990) *Time Series Analysis: Univariate and Multivariate Methods* Addison-Wesley

5 Parameters

- 1:** K — INTEGER *Input*
On entry: the dimension, k , of the multivariate time series.
Constraint: $K \geq 1$.
- 2:** IP — INTEGER *Input*
On entry: the number of AR (or MA) parameter matrices, p (or q).
Constraint: $IP \geq 1$.
- 3:** PAR(IP*K*K) — *real* array *Input*
On entry: the AR (or MA) parameter matrices read in row by row in the order $\phi_1, \phi_2, \dots, \phi_p$ (or $\theta_1, \theta_2, \dots, \theta_q$). That is PAR($(l-1) \times k \times k + (i-1) \times k + j$) must be set equal to the (i, j) th element of ϕ_l , for $l = 1, 2, \dots, p$ (or the (i, j) th element of θ_l , for $l = 1, 2, \dots, q$).
- 4:** RR(K*IP) — *real* array *Output*
On exit: the real parts of the eigenvalues.
- 5:** RI(K*IP) — *real* array *Output*
On exit: the imaginary parts of the eigenvalues.
- 6:** RMOD(K*IP) — *real* array *Output*
On exit: the moduli of the eigenvalues.
- 7:** WORK(K*K*IP*IP) — *real* array *Workspace*
- 8:** IWORK(K*IP) — INTEGER array *Workspace*
- 9:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, $K < 1$,
 or $IP < 1$.

IFAIL = 2

An excessive number of iterations are needed to evaluate the eigenvalues of $A(\phi)$ (or $B(\theta)$). This is an unlikely exit. All output parameters are undefined.

7 Accuracy

The accuracy of the results depends on the original matrix and the multiplicity of the roots.

8 Further Comments

The time taken is approximately proportional to kp^3 (or kq^3).

9 Example

This example program finds the eigenvalues of $A(\phi)$ where $k = 2$ and $p = 1$ and $\phi_1 = \begin{bmatrix} 0.802 & 0.065 \\ 0.000 & 0.575 \end{bmatrix}$.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G13DXF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          KMAX, IPMAX
      PARAMETER        (KMAX=6,IPMAX=3)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, IP, K, NPAR
*      .. Local Arrays ..
      real            PAR(KMAX*KMAX*IPMAX), RI(KMAX*IPMAX),
+                   RMOD(KMAX*IPMAX), RR(KMAX*IPMAX),
+                   WORK(KMAX*KMAX*IPMAX*IPMAX)
      INTEGER          IW(KMAX*IPMAX)
*      .. External Subroutines ..
      EXTERNAL         G13DXF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G13DXF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) K, IP
      IF (K.GT.0 .AND. K.LE.KMAX .AND. IP.GT.0 .AND. IP.LE.IPMAX) THEN
*         Read the AR (or MA) parameters
         NPAR = IP*K*K
         READ (NIN,*) (PAR(I),I=1,NPAR)
         IFAIL = 0
*
         CALL G13DXF(K,IP,PAR,RR,RI,RMOD,WORK,IW,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) '          Eigenvalues          Moduli'
         WRITE (NOUT,*) '          -----          -----'
*
         DO 20 I = 1, K*IP
            IF (RI(I).GE.0.0e0) THEN
               WRITE (NOUT,99999) RR(I), RI(I), RMOD(I)
            ELSE
               WRITE (NOUT,99998) RR(I), -RI(I), RMOD(I)
            END IF
20      CONTINUE
         ELSE
            WRITE (NOUT,*) ' Either K or IP is out of range'
         END IF
         STOP
*
99999 FORMAT (' ',F10.3,' + ',F6.3,' i ',F8.3)
99998 FORMAT (' ',F10.3,' - ',F6.3,' i ',F8.3)
      END

```

9.2 Program Data

G13DXF Example Program Data

```
2 1
0.802 0.065
0.000 0.575
```

9.3 Program Results

G13DXF Example Program Results

Eigenvalues	Moduli
-----	-----
0.802 + 0.000 i	0.802
0.575 + 0.000 i	0.575
